# $Id: PKGBUILD 87450 2013-03-31 20:44:06Z alucryd $
# Maintainer: Maxime Gauduin <alucryd@gmail.com>

pkgname=python2-slip
pkgver=0.4.0
pkgrel=1
pkgdesc="Simple Library for Python"
arch=('any')
url="http://fedorahosted.org/python-slip"
license=('GPL2')
depends=('gtk2' 'polkit' 'pygtk' 'python2-dbus' 'python2-decorator' 'python2-gobject')
source=("http://fedorahosted.org/released/python-slip/python-slip-${pkgver}.tar.bz2")
sha256sums=('db5afa59681e17f59559c5e1cc32447ae138120ea3058111dd5c61745e079212')

build() {
  cd "${srcdir}"/python-slip-${pkgver}

# Patch
  sed -i 's|^#!.*python$|#!/usr/bin/python2|' $(grep -rl '^#!.*python')
  sed -i 's|python|python2|g' py_rules.mk

# Build
  make
}

package() {
  cd "${srcdir}"/python-slip-${pkgver}

# Install
  make DESTDIR="${pkgdir}" install
}

# vim: ts=2 sw=2 et:
